
Tópicos abordados 

Estruturas de Decisão 

 Simples 

 Composta 

 Encadeada 



Estruturas Condicionais 

ou de Decisão 
Permite a escolha de um grupo de ações para serem 
executadas de acordo com a aceitação ou não de certas 
condições.  

São testados parâmetros e, a depender de seus valores, 
tomamos um caminho ou outro.  

As condições que são testadas num Algoritmo são do 
tipo lógica (booleana), portanto podem ter somente dois 
resultados: Verdadeiro ou Falso. Assim, a seleção de 
ações pode seguir, no máximo, duas alternativas: uma 
se a condição for verdadeira e outra se a condição 
testada for falsa. Existem seleções mais simples e 
seleções compostas outras seleções.  
 



Sintaxe 

 

 
Observação: as palavras sublinhadas na Sintaxe são 

chamadas de “palavras reservadas” e sempre aparecem 

na estrutura. As palavras em itálico são indicações do tipo 

de expressão ou comando que devem ser usados.  

 

Exemplo: se (Media_final>=5) então 

        escreva (“Aluno Aprovado!”) 

          fimse 

Estruturas Condicionais 

ou de Decisão Simples 



Estruturas Condicionais 

ou de Decisão Composta 

 

 

 
 

Exemplo: 

    se (Media_final>=5) então 

         escreva (“Aluno Aprovado!”) 

    senão 

                  escreva (“Aluno Reprovado!”) 

   fimse 



Estruturas Condicionais ou de 

Decisão Aninhadas ou Encadeadas 

 

 

 

 

 
Exemplo: 
se (Media_final>=5) então 
         escreva (“Aluno Aprovado!”) 
senão 
                  se (Media_Final>=2.5) então 
                       escreva (“Aluno de Recuperação!”) 
                  senão 
                        escreva (“Aluno Reprovado!”) 
                  fimse 
fimse 



Exemplos: Condicional Simples 
algoritmo "MediaAluno" 

// Função : Calcular a média de um aluno e dizer se ele está aprovado ou não 

// Autor : Sérgio Carlos Portari Junior 

// Data : 11/3/2011 

var 

   nome_aluno : caracter 

   n1,n2,n3,n4 : real 

   soma : real 

   media : real 

inicio 

      escreva("Digite o Nome do Aluno: ") 

      leia(nome_aluno) 

      escreva("Digite a primeira nota: ") 

      leia(n1) 

      escreva("Digite a segunda nota: ") 

      leia(n2) 

      escreva("Digite a terceira nota: ") 

      leia(n3) 

      escreva("Digite a quarta nota: ") 

      leia(n4) 

      soma :=(n1+n2+n3+n4) 

      media:=(soma/4) 

      escreva(media) 

      se (media>=5) entao 

         escreva (" - Aluno Aprovado") 

      fimse 
fimalgoritmo 



Exemplos: Condicional Composta  
algoritmo "MediaAluno" 

// Função : Calcular a média de um aluno e dizer se ele está aprovado ou não 

// Autor : Sérgio Carlos Portari Junior 

// Data : 11/3/2011 

var 

   nome_aluno : caracter 

   n1,n2,n3,n4 : real 

   soma : real 

   media : real 

inicio 

      escreva("Digite o Nome do Aluno: ") 

      leia(nome_aluno) 

      escreva("Digite a primeira nota: ") 

      leia(n1) 

      escreva("Digite a segunda nota: ") 

      leia(n2) 

      escreva("Digite a terceira nota: ") 

      leia(n3) 

      escreva("Digite a quarta nota: ") 

      leia(n4) 

      soma :=(n1+n2+n3+n4) 

      media:=(soma/4) 

      escreva(media) 

      se (media>=5) entao 

         escreva (" - Aluno Aprovado") 

      senao 

         escreva (" - Aluno Reprovado") 

      fimse 
fimalgoritmo 



Exemplos: Condicional Encadeada 
algoritmo "MediaAluno" 

// Função : Calcular a média de um aluno e dizer se ele está aprovado ou não 

// Autor : Sérgio Carlos Portari Junior 

// Data : 11/3/2011 

var 

   nome_aluno : caracter 

   n1,n2,n3,n4 : real 

   soma : real 

   media : real 

inicio 

      escreva("Digite o Nome do Aluno: ") 

      leia(nome_aluno) 

      escreva("Digite a primeira nota: ") 

      leia(n1) 

      escreva("Digite a segunda nota: ") 

      leia(n2) 

      escreva("Digite a terceira nota: ") 

      leia(n3) 

      escreva("Digite a quarta nota: ") 

      leia(n4) 

      soma :=(n1+n2+n3+n4) 

      media:=(soma/4) 

      escreva(media) 

      se (media>=5) entao 

         escreva (" - Aluno Aprovado") 

      senao 

          se (media>=2.5) entao 

               escreva (“ – Aluno de Recuperação!”) 

          senao 

               escreva (" - Aluno Reprovado") 

          fimse 

      fimse 
fimalgoritmo 



Exemplo 

Dados três valores (A,B e C),  verificar se eles podem 
ser os comprimentos dos lados de um triângulo. Se 
forem, verificar se compõe um triângulo equilátero, 
isóceles ou escaleno. Informar se não compuserem um 
triângulo. 

Dados de entrada? 
 Valores para A, B e C 

Processamento? 
 Comparação dos valores e condições de composição de um 

triângulo 

Dados de Saída? 
 Mensagens: É um triângulo equilátero, É um triângulo isóceles, 

É um triângulo escaleno, Não é um triângulo 



Exemplo 

Observação: Para resolver este problema, 
precisamos saber claramente: 

O que é um triângulo? 
 Uma figura geométrica fechada, de três lados, em 

que cada lado é menor do que a soma dos outros 
dois lados. 

O que é um triângulo eqüilátero? 
 Um triângulo com os três lados iguais 

O que é um triângulo isóceles? 
 Um triângulo com dois lados iguais 

O que é um triângulo escaleno? 
 Um triângulo com todos os lados diferentes 



Exemplo 

Formulando as condições 
1. Saber se é um triângulo: 

(A<B+C) E (B<A+C) E (C<A+B) 

2. Saber se é um triângulo equilátero: 

(A=B) E (B=C) 

3. Saber se é um triângulo isóceles: 

(A=B) OU (A=C) OU (B=C) 

4. Saber se é um triângulo escaleno: 

(A<>B) E (B<>C) E (A<>C) 



Exemplo 
 

Primeiro passo: Criar as variáveis que serão informadas para 
análise  

     A,B,C : inteiro 

 

Segundo passo: Criar os comandos para a entrada dos dados, ou 
seja, ler os valores para A, B e C 

  //entrada de dados 

  escreva ("Digite o valor de A: ") 

  leia (A) 

  escreva ("Digite o valor de B: ") 

  leia (B) 

  escreva ("Digite o valor de C: ") 

  leia (C) 

  //fim da entrada de dados 

 

Terceiro passo: Descobrir se os lados A,B,C informados compõe 
um triângulo. Para isso,  usaremos uma condicional composta 
combinada com operadores lógicos. 



Exemplo 
se ((A<B+C) e (B<A+C) e (C<A+B)) então 

Feito isso, veremos as opções possíveis (verdadeiro e falso) 

Quarto passo: Resultado verdadeiro (então é um triângulo),  

iremos ver se o triângulo então é eqüilátero. 

 se ((A=B) e (B=C)) então 

Temos agora mais dois resultados (verdadeiro e falso) 

Quinto passo: Resultado verdadeiro (então é eqüilátero), iremos 

informar a mensagem que o triângulo é eqüilátero (saída) 

  escreva (“É um triângulo eqüilátero”) 

Sexto passo: Finalizamos a opção verdadeira, veremos a opção 

falsa da condição anterior. Se não for eqüilátero, ele ainda 

pode ser isóceles. 

 senão 

 se ((A=B) ou (B=C) ou (A=C)) então 

  escreva (“É um triângulo isóceles”) 



Exemplo 
Finalizamos a opção verdadeira do isóceles, então nosso triângulo só 

pode ser escaleno, sem precisar nem ver a última condição, pois 
verificamos se os lados constituíam um triângulo, verificamos se os 
três lados eram iguais, verificamos se dois lados eram iguais e 
chegamos na única opção restante, que constitui três lados 
diferentes. 

 senão 

  escreva (“É um triângulo escaleno”) 

Temos agora apenas que exibir a mensagem da primeira condicional, 
informando que os lados não constituem um triângulo. Para isso, 
precisamos fechar as duas condicionais abertas em isóceles e 
eqüilátero e fechar o algoritmo, que representam as condições falsas 
que não precisarão ser analizados. 

  fimse 

 fimse 

 senão 

  escreva (“Não é um triângulo”) 

 fimse 

fimalgoritmo 



Exemplo 
algoritmo "triângulo" 

// Função : Descobrir se três lados formam um triângulo e que tipo de triângulo 

// Autor :  Sérgio Carlos Portari Jr 

// Data : 11/03/2011 

var 

  A,B,C : inteiro 

inicio 

escreva ("Digite o valor de A: ") 

leia (A) 

escreva ("Digite o valor de B: ") 

leia (B) 

escreva ("Digite o valor de C: ") 

leia (C) 

se (A<B+C) e (B<A+C) e (C<A+B) entao 

    se ((A=B) e (B=C)) entao 

       escreva ("É um triângulo eqüilátero") 

    senao 

       se ((A=B) ou (B=C) ou (A=C)) entao 

        escreva ("É um triângulo isóceles") 

         senao 

        escreva ("É um triângulo escaleno") 

         fimse 

    fimse 

senao 

  escreva ("Não é um triângulo") 

fimse 

fimalgoritmo 



Exercícios 

1. Faça um algoritmo que leia a idade de 
uma pessoa e mostre uma mensagem 
dizendo se essa pessoa é maior ou 
menor de 18 anos. 

 

2. Faça um algoritmo que receba dois 
números inteiros e verifique se o 
primeiro é maior que o segundo ou se o 
segundo é maior que o primeiro e mostre 
na tela qual é o maior. 



Exercícios 

3. Escreva um algoritmo que receba três 

números reais diferentes e mostre-os em 

ordem crescente (do menor para o maior) 

 

4. Elabore um algoritmo que leia dois números 

inteiros e um dos sinais de operação +,-,* ou /. 

Depois de ler os dados, realize a operação do 

segundo número pelo primeiro segundo o sinal 

de operação lido, sabendo que o segundo 

número não pode ser zero. 


